Sommation symbolique

École Jeunes Chercheurs en Algorithmique et Calcul formel 2003

Frédéric Chyzak

  frederic.chyzak@inria.fr
http://algo.inria.fr/chyzak/

Suites et sommes

Une suite (u_n)_{n∈N} associe une valeur à chaque entier n. La série de terme général u_n, notée ∑ k≥0 u_k, est la suite (U_n)_{n∈N} des sommes partielles

U_n = u_0 + · · · + u_n = n∑ k=0 u_k.

La série converge lorsque U_n a une limite S, notée alors ∞∑ k=0 u_k.

U_n est une somme indéfinie de u_n ; elle vérifie U_n − U_{n−1} = u_n. La somme S de la série est la somme définie des u_n sur N.

Séries génératrices

La série génératrice de la suite (u_n)_{n∈N} ∈ C^N est la série formelle U = ∑ n≥0 u_nX^n ∈ C [[X]].

U est toujours convergente dans C [[X]].

Si U converge au sens des fonctions (sur un voisinage complexe de 0), on identifie la série entière et la fonction analytique U(z) = ∞∑ n=0 u_nz^n.

Sommation symbolique : sommes indéfinies

Si u_n est donné sous forme close :

• polynomiale, u_n = P (n) pour P ∈ C [X], • rationnelle, u_n = R(n) pour R ∈ C (X), • hypergéométrique, u_{n+1}/u_n = R(n) pour R ∈ C (X), • q-hypergéométrique, u_{n+1}/u_n = R(q^n) pour R ∈ C (q, X), • polynomialement récursive,

P_d(n)u_{n+d} + · · · + P_0(n)u_n = 0 pour des P_i ∈ C [X], • par des sommes emboîtées selon certaines contraintes (alembertienne, liouvillienne, etc), déterminer s’il existe une forme close U_n telle que U_n − U_{n−1} = u_n.

Exemple : sommation indéfinie hypergéométrique n∑ k=0 4^k (_{2k} k ) = 2(n + 1) 3 4^n (_{2n} n ) + 1 3 .

b_{n,m} = ( n m ) = coefficient binomial = n! m! (n − m)! ⇒ b_{n+1,m} b_{n,m} = n + 1 n + 1 − m , b_{n,m+1} b_{n,m} = n − m m + 1 .

u_n =

4^n (_{2n} n ) ⇒ u_{n+1} u_n = 2(n + 1) 2n + 1 .

Exemple : sommation indéfinie polynomialement récursive n−1∑ k=p+1 ( k p ) H_k = ( (n + 1)^2 (p + 1)^2 − 1 ) ( n p ) H_n − (n − p)(n − p + 1) (p + 1)^2 ( n + 1 p ) H_{n+1}

= ( n p ) n − p (p + 1)^2 ( (p + 1)H_n − 1 ) .

H_n = nombre harmonique = 1 + 1 2 + 1 3 + · · · + 1 n , u_n = ( n p ) H_n ⇒ (n + 1 − p)(n + 2 − p)u_{n+2} − (2n + 3)(n + 1 − p)u_{n+1} + (n + 1)^2u_n = 0.

Sommation symbolique : sommes définies

Constantes : difficile, peu de choses à dire (sur la sommation). ∞∑ n=1 1 n^s = ζ(s), ζ(2p) ∈ Q π^{2p} .

Sommes paramétrées : si u est donné sous forme close (hypergéométrique, q-hypergéométrique, ∂-finie), déterminer respectivement

U_n = ∞∑ k=0 u_{n,k} ou U(z) = ∞∑ k=0 u_k(z) sous forme close.

Exemples : sommation définie (q-)hypergéométrique

2n∑ k=0 (−1)^k ( 2n k )( 2k k )( 4n − 2k 2n − k ) = ( 2n n )_2 , +∞∑ −∞ (a; q)_nx^n (b; q)_n = (ax; q)_∞ (_q ax ; q ) ∞ (q; q)_∞ (_b a ; q ) ∞ (x; q)_∞ (_b ax ; q ) ∞ (b; q)_∞ (_q a ; q ) ∞ (Ramanujan).

(x; q)_n = q-analogue du symbole de Pochhammer = (1 − x)(1 − qx) . . . (1 − q^{n−1}x) ⇒ (x; q)_{n+1} (x; q)_n = 1 − q^nx ∈ F, (qx; q)_n (x; q)_n = 1 − q^nx 1 − x ∈ F. pour F = C (q, x, q^n ).

Exemple : sommation définie ∂-finie (1)

n∑ k=0 (_k∑ j=0 ( n j ))_3 = n2^{3n−1} + 2^{3n} − 3n2^{n−2} ( 2n n ) (Calkin).

v_{n,k} = k∑ j=0 ( n j ) ⇒ les v_{n+i,k+j} se récrivent sur (v_{n,k}, v_{n,k+1}) :

(k + 2)v_{n,k+2} − (n + 1)v_{n,k+1} + (n − k − 1)v_{n,k} = 0, (k + 1)v_{n,k+1} + (n − k)v_{n+1,k} − (2n + 1 − k)v_{n,k} = 0. u_{n,k} = v^3_{n,k} ⇒ les u_{n+i,k+j} se récrivent sur (u_{n,k}, u_{n+1,k}, u_{n,k+1}, u_{n,k+2}).

Exemple : sommation définie ∂-finie (2) +∞∑ k=−∞ J_k(z)^2 = 1 (Neumann).

J_k(z) = fonctions de Bessel = (z/2)^k ∞∑ n=0 (−z^2/4)^n n! (n + k)! ⇒ les J^{(j)}_{k+i}(z) se récrivent sur ( J_k(z), J^′_k(z) ) ou sur ( J_k(z), J_{k+1}(z) ) : z^2J^{′′}_k (z)+zJ^′_k (z)+(z^2−k^2)J_k(z) = 0, zJ^′_k(z)+zJ_{k+1}(z)−kJ_k(z) = 0, zJ_{k+2}(z) − 2(k + 1)J_{k+1}(z) + zJ_k(z) = 0.

u_k(z) = J_k(z)^2 ⇒

les u^{(j)}_{k+1}(z) se récrivent sur ( u_k(z), u_{k+1}(z), u^′_k(z) ) .

Algorithmes et méthodes principaux

1. Séries génératrices

• Conversion entre suites de coefficients et séries génératrices 2. Suites hypergéométriques :

• Méthode de Fasenmyer pour les sommes définies • Algorithme de Gosper pour les sommes indéfinies • Algorithme de Zeilberger pour les sommes définies 3. Suites ∂-finies :

• Bases de Gröbner pour les sommes indéfinies et définies • Algorithmes de Chyzak pour les sommes indéfinies et définies 4. Sommes dans des tours de corps à différences • Algorithme de Karr pour les sommes indéfinies

Séries génératrices : le cas différentiellement finies

Méthode : Soit (f_n)_{n≥0} vérifiant une relation de récurrence linéaire à coefficients polynomiaux en n. Déduire une équation différentielle linéaire à coefficients polynomiaux en u dont la série génératrice F (z) = ∑ n≥0 f_nz^n est solution. Résoudre.

Exemple : pour les polynômes orthogonaux de Hermite H_n(z), ∑ n≥0 H_n(u)H_n(v)z^n n! = e^{4z uv−(u} 2+v2)z 1−4z2 √ 1 − 4z^2 .

with(Mgfun):
expr:=HermiteH(n,u)*HermiteH(n,v)/n!:
rec:=dfinite_expr_to_rec(expr,f(n));

rec := { (16n+16)f(n)−16uvf(n+1)+(8u^2 +8v^2 −8n−20)f(n+2)

− 4uvf(n + 3) + (n + 4)f(n + 4) }

init_cond:={seq(f(i)=
    eval(subs(n=i,HermiteH=orthopoly[H],expr)),i=0..3)};

init cond := { f(0) = 1, f(1) = 4uv, f(2) = 1 2 (4u^2 − 2)(4v^2 − 2),

f(3) = 1 6 (8u^3 − 12u)(8v^3 − 12v) }

with(gfun):
deq:=rectodiffeq(rec union init_cond,f(n),F(z));

deq := { F (0) = 1,

(16z^3 −16z^2uv+8zu^2 +8zv^2 −4z−4uv)F (z)+(16z^4 −8z^2 +1)F^′(z) }

dsolve(deq);

F (z) = e^{u2+v2−4uvz} (2z+1)(2z−1) eu 2 e^{v2} I √ 2z + 1 √ 2z − 1

Transformation de Mellin formelle ∑ n≥0 f_nz^n = F (z), f_n = 0 si n < 0 ⇒ ∑ n≥0 nf_nz^n = zF^′(z), ∑ n≥0 f_{n−1}z^n = zF (z)

n ↔ z∂, σ ↔ z σ = décalage arrière, ∂ = dérivation σn = (n − 1)σ, ∂z = z∂ + 1 P (n, σ)(f)(n) = 0 ↔ P (z∂, z)(F )(z) = 0 si récurrence valable sur Z !

Creative Telescoping

Soit à évaluer la somme

U_n = b∑ k=a u_{n,k} a ∈ Z ∪ {−∞}, b ∈ Z ∪ {+∞}. Supposons que l’on détermine une relation de la forme η_d(n)u_{n+d,k} + · · · + η_0(n)u_{n,k} = v_{n,k+1} − v_{n,k}. Alors, après sommation sur k entre a et b, on trouve η_d(n)U_{n+d} + · · · + η_0(n)U_n = v_{n,b+1} − v_{n,a}.

Sommation à bornes naturelles : quand v_{n,a} = v_{n,b+1}.

Suites hypergéométriques

Une suite f = (f_{n,k}) est dite hypergéométrique si f_{n−1,k} f_{n,k} ∈ C (n, k) et f_{n,k−1} f_{n,k} ∈ C (n, k).

Une suite f = (f_{n,k}) est dite q-hypergéométrique si f_{n−1,k} f_{n,k} ∈ C (q, q^n , q^k) et f_{n,k−1} f_{n,k} ∈ C (q, q^n , q^k).

Exemples : produits et quotients de factorielle et de coefficients binomiaux et multinomiaux ; factorielles montantes et descendantes ; q-analogues.

La méthode de Fasenmyer

Entrée : terme hypergéométrique f_{n,k}.

Sortie : récurrence linéaire sur f à coefficients polynomiaux ne faisant pas intervenir k.

Méthode :

1. Pour un ensemble de décalages S donné, on cherche à résoudre ∑ (i,j)∈S c_{i,j} (n)f_{n−i,k−j} = 0 ;

2. Après division par f_{n,k} et après avoir chassé les dénominateurs, on trouve des polynômes p_{i,j} tels que ∑ (i,j)∈S c_{i,j} (n)p_{i,j}(n, k) = 0 ;

3. Égaler les coefficients des k^l à 0 et résoudre le système linéaire en les c_{i,j} ainsi obtenus ;

4. On obtient ∑ (i,j)∈S c_{i,j} (n)f_{n−i,k} = g_{n,k} − g_{n,k−1} pour un g adéquat ;

5. Après sommation en k et introduction de F_n = b∑ k=a f_{n,k}, ∑ (i,j)∈S c_{i,j} (n)F_{n−i} = g_{n,b} − g_{n,a−1}.

Améliorations : optimisation de S par Verbaeten ; multisommes et dépendance polynomiale des c_{i,j} en k par Wegschaider ; q-analogues par Riese.

Identité de Dixon

Sous la forme : n∑ k=0 (−1)^k ( n k )_3 = (−1)^p (_{3p} p,p,p ) si n = 2p, 0 si n = 2p + 1.

Pour S = {0, 1, 2, 3}^2, la méthode de Fasenmyer renvoie (rapidement) un résultat positif, conduisant à la relation

3(3n − 2)(3n − 4)F_{n−2} + n^2F_n = 0.

sommant:=(-1)^k*binomial(n,k)^3: N:=3:
P:=collect(numer(normal(add(add(c[i,j]*normal(
    subs([n=n-i,k=k-j],sommant)/sommant,expanded),
j=0..N),i=0..N))),k): sys:={coeffs(P,k)}:
sol:=solve(sys,{seq(seq(c[i,j],j=0..N),i=0..N)}):
map(factor,subs(sol,c[0,0]=n^2*(3*n-5),
    add(add(c[i,j]*f(n-i,k-j),j=0..N),i=0..N)))=0;

n^2(3n − 5)f_{n,k} + (9n^3 − 24n^2 + 17n − 4)f_{n−1,k−1} + (3n − 4)(21n^2 − 49n + 24)f_{n−2,k−1} + (3n − 4)(3n^2 − 7n + 3)f_{n−2,k−2} + 3(3n − 2)(n − 2)^2f_{n−3,k−1} − 3(3n − 2)(n − 2)^2f_{n−3,k−2} + (3n − 2)(n − 2)^2f_{n−3,k−3} + (−9n^3 + 24n^2 − 17n + 4)f_{n−1,k} + (3n − 4)(3n^2 − 7n + 3)f_{n−2,k} − (3n − 2)(n − 2)^2f_{n−3,k} = 0

rec:=collect(map(factor,subs(sol,c[0,0]=n^2,
    add(add(c[i,j]*F(n-i),j=0..N),i=0..N))),F,factor)=0;

rec := 3(3n − 2)(3n − 4)F_{n−2} + n^2F_n = 0

LREtools[hypergeomsols](eval(subs(n=2*p,
    F=unapply(G(n/2),n),rec)),G(p),{},output=basis);

C_1(−3)^p(3^p)^2Γ(p + 2/3)Γ(p + 1/3) √ 3 2Γ(p + 1)^2π remove(has,%,GAMMA)*convert(GAMMA(3*p)/GAMMA(p)*
    simplify(select(has,%,GAMMA)*GAMMA(p)/GAMMA(3*p)),
    factorial):
normal(simplify(remove(has,%,factorial))*normal(
    select(has,%,factorial),expanded));

(3p)! (−1)^p C_1 p!^3

seq(f[i]=add(eval(subs([n=i,k=j],sommant)),j=0..i),
    i=0..6);

f_0 = 1, f_1 = 0, f_2 = −6, f_3 = 0, f_4 = 90, f_5 = 0, f_6 = −1680

Autres exemples typiques par cette méthode ∑ i,j ( i + j i )_2( 4n − 2i − 2j 2n − 2i ) = (2n + 1) ( 2n n )_2 (Andrews–Paule) ∑ j,k (−1)^{j+k} ( j + k k + l )( r j )( n k )( s + n − j − k m − j )

= (−1)^l ( n + r n + l )( s − r m − n − l )

(Graham–Knuth–Patashnik) ∑ r,s (−a)^{n+r+s} ( n r )( n s )( n + s s )( n + r r )( 2n − r − s n ) = ∑ k ( n k )_4

(Petkovšek–Wilf–Zeilberger)

Algorithme de Gosper

Entrée : terme hypergéométrique f_k.

Sortie : fraction rationnelle R(k) telle que Rf soit une somme indéfinie en k de f, ou une preuve qu’aucune telle R n’existe.

Vue simplifiée de l’algorithme : L’équation

f_k = R(k + 1)f_{k+1} − R(k)f_k

en une fraction rationnelle R s’écrit encore 1 = R(k + 1)ρ(k) − R(k), équation qui se résout par l’algorithme de décision d’Abramov : si aucune R n’est trouvée, on a une preuve qu’aucune n’existe.

Algorithme original :

1. Calculer le rapport ρ(k) = f_{k+1}/f_k et le représenter sous la forme (unique) ρ(k) = p(k + 1) p(k) q(k) r(k + 1) de façon que pgcd(p, q) = pgcd(p, r) = pgcd ( q(k), r(k + h) ) = 1 pour tout entier h > 0.

2. Le changement de variable R = rS/p ramène l’équation en la fraction rationnelle R en une équation en un polynôme S : p(k) = q(k)S(k + 1) − r(k)S(k). 3. Résoudre (des bornes sur le degré existent). Si un S est trouvé, renvoyer R = rS/p ; sinon, on a une preuve qu’aucune R n’existe.

Exemple d’exécution de l’algorithme de Gosper

n−1∑ k=0 (−1)^k(4k + 1)

(_{2k+1} k ) 4^k(4k^2 − 1) = − 2(n + 1) 4n + 1

(−1)^n(4n + 1)

(_{2n+1} n ) 4^n(4n^2 − 1) − 2.

En Maple8 :

f:=(-1)^k*(4*k+1)/4^k/(4*k^2-1)*binomial(2*k+1,k):
SumTools:-Hypergeometric:-IndefiniteSum(f,k); %/f;

− 2(k + 1)(−1)^k binomial(2k + 1, k) 4^k(4k^2 − 1) − 2(k + 1) 4k + 1

Par résolution de l’équation rationnelle par l’algorithme d’Abramov :

rho:=factor(normal(subs(k=k+1,f)/f,expanded));

ρ := − 1 2 (4k + 5)(2k − 1) (4k + 1)(k + 2)

LREtools[ratpolysols](1=R(k+1)*rho-R(k),R(k),
    {},output=basis);

− 2(k + 1) 4k + 1

Par construction et résolution de l’équation polynomiale :

p:=gcd(subs(k=k-1,numer(rho)),denom(rho));
rho/(subs(k=k+1,p)/p);

p := 4k + 1 − 1 2 2k − 1 k + 2

q:=numer(%); r:=subs(k=k-1,denom(%%));
solve(resultant(q,subs(k=k+h,r),k),h);

q := −2k + 1 r := 2k + 2 −3 2

LREtools[polysols](p=q*S(k+1)-r*S(k),S(k),
    {},output=basis);

−1

r*%/p;

− 2(k + 1) 4k + 1

Algorithme de Gosper paramétré

Entrée : un terme hypergéométrique f_k et des fractions rationnelles s_0(k), . . . , s_m(k).

Sortie : une fraction rationnelle R(k) et des constantes η_0, . . . , η_m telles que Rf soit une somme indéfinie en k de ( η_0s_0(k) + · · · + η_ms_m(k) ) f_k, ou une preuve qu’aucune telle R n’existe pour aucune famille {η_i}.

Idée de l’algorithme : Les paramètres η_i n’interviennent que de façon linéaire et au second membre de l’équation polynomiale en S(k). Le calcul de S se résume à la résolution d’un système linéaire lors de laquelle on peut prendre en compte les inconnues additionnelles η_0, . . . , η_m.

Algorithme de Zeilberger

Entrée : terme hypergéométrique f_{n,k}.

Sortie : fractions rationnelles η_0(n), . . . , η_d(n), ϕ(n, k) pour d minimal telles que η_d(n)f_{n+d,k} + . . . η_0(n)f_{n,k} = ϕ(n, k + 1)f_{n,k+1} − ϕ(n, k)f_{n,k}.

Terminaison non garantie en général ; l’est dans le cas holonome. Critère récent d’Abramov.

Algorithme : Calculer les fractions rationnelles s_i(n, k) telles que f_{n+i,k} = s_i(n, k)f_{n,k} et utiliser l’algorithme de Gosper paramétré.

Exemple d’exécution de l’algorithme de Zeilberger

Les polynômes orthogonaux de Jacobi P^{(α,β)}n (x) s’expriment en termes de la fonction hypergéométrique de Gauss_2F_1 (_{a,b} c | |z) : P^{(α,β)} n (x) =

(α + 1)_n

n!^2F^1 ( −n, n + α + β + 1 α + 1 | | | 1 − x 2 ) , pour_2F_1 (_{a,b} c | |z) = ∑^∞_{k=0}^{(a)k (b)k}_{(c)k k!} z^k où (x)_k = x(x + 1) . . . (x + k − 1).

L’algorithme fournit la récurrence en n sur P^{(α,β)}_n (x),

0 = 2(n + 2)(n + α + β + 2)(2n + α + β + 2)P^{(α,β)}_{n+2} (x) − ( (2n + α + β + 2)3x + (2n + α + β + 3)(α − β)(α + β) ) P^{(α,β)} n+1 (x)

+ 2(n + α + 1)(n + β + 1)(2n + α + β + 4)P^{(α,β)}_n (x),

et fournirait toutes les relations de contiguïté.

f:=pochhammer(alpha+1,n)/n!
    *pochhammer(-n,k)*pochhammer(n+alpha+beta+1,k)
    /pochhammer(alpha+1,k)/k!*(1-x)^k/2^k:
rho:=factor(normal(subs(k=k+1,f)/f,expanded));

ρ := 1 2 (n − k)(n + α + β + 1 + k)(−1 + x) (k + 1)(α + 1 + k)

rho*R(k+1)-R(k)-add(eta[i]*factor(
    normal(subs(n=n+i,f)/f,expanded)),i=0..2):
rec:=collect(numer(normal(%)),
    [R,seq(eta[i],i=0..2)],factor);

rec := (n + 2 − k)(n + 1 − k)(n − k)(n + 2 + α + β) (n + α + β + 1)(n + α + β + 1 + k)(−1 + x)R(k + 1) − 2(k + 1)(α + 1 + k)(n + α + β + 1) (n + 1 − k)(n + 2 − k)(n + 2 + α + β)R(k) − 2η_0(k + 1)(α + 1 + k)(n + α + β + 1) (n + 1 − k)(n + 2 − k)(n + 2 + α + β) − 2η_1(n + α + β + 1 + k)(α + 1 + n)(k + 1) (α + 1 + k)(n + 2 − k)(n + 2 + α + β) − 2η_2(α + 2 + n)(α + 1 + n)(n + α + β + 2 + k) (n + α + β + 1 + k)(k + 1)(α + 1 + k)

den:=primpart(gcd(subs(k=k-1,coeff(rec,R(k+1))),
    coeff(rec,R(k))),k);

den := (n + 2 − k)(n + 1 − k)

REC:=collect(primpart(eval(subs(R=unapply(S(k)/den,k),
    rec)),S(k+1)),[S,seq(eta[i],i=0..2)],factor):
collect(REC,[S,seq(eta[i],i=0..2)],p->degree(p,k));

2S(k + 1) + 2S(k) + 4η_0 + 4η_1 + 4η_2

lcoeff(coeff(REC,S(k)),k),lcoeff(coeff(REC,S(k+1)),k);

−2(n+2+α+β)(n+α+β +1), −(n+2+α+β)(n+α+β +1)(−1+x) num:=add(c[i]*k^i,i=0..2): sys:={coeffs(collect(
    eval(subs(S=unapply(num,k),REC)),k,normal),k)}:
sol:=solve(sys,{seq(c[i],i=0..2),seq(eta[i],i=0..2)}):
P:=collect(subs(sol,add(eta[i]*u(n+i,k),i=0..2)),
    u,factor): Q:=factor(subs(sol,num/den))*u(n,k):
denom(P): P:=subs(eta[0]=%,P); Q:=subs(eta[0]=%%,Q);

P := 2(β + n + 1)(α + 1 + n)(α + 2n + 4 + β)u(n, k) − (3 + α + β + 2n)(α^2 + α^2x + 4αxn + 2αβx + 6αx + 12xn + β^2x + 4xnβ + 4xn^2 + 6βx + 8x − β^2)u(n + 1, k) + 2(n + 2)(n + 2 + α + β)(α + 2 + 2n + β)u(n + 2, k) Q := −2k(3 + α + β + 2n)(α + 2 + 2n + β)(α + 1 + n)(α + k) (α + 2n + 4 + β)/(n + 1 − k)/(n + 2 − k)/(n + α + β + 1)u(n, k)

Suites ∂-finies

Généralisation des suites polynomialement récursives à un indice.

Un terme u_{n,k} est dit ∂-fini lorsque l’ensemble de ses décalés u_{n+i,k+j} engendre un espace vectoriel de dimension finie sur C (n, k).

Un terme u_k(z) est dit ∂-fini lorsque l’ensemble de ses dérivées et décalés u^{(i)}_{k+j} (z) engendre un espace vectoriel de dimension finie sur C (k, z).

Exemple : familles de polynômes orthogonaux classiques, fonctions de Bessel, fonctions hypergéométriques et généralisations, sommes indéfinies de suites hypergéométriques, etc, l’algèbre de ces familles.

Algorithme de Gosper étendu

Entrée : un espace vectoriel V de base ( b_1(k), . . . , b_r(k) ) stable par décalage en k ; un terme ∂-fini f_k de V .

Sortie : des fractions rationnelles ϕ_1(k), . . . , ϕ_r(k) telles que ϕ_1b_1 + · · · + ϕ_rb_r soit une somme indéfinie en k de f, ou une preuve que de telles ϕ_i n’existent pas. Algorithme :

1. Poser F_k = ϕ_1(k)b_1(k) + · · · + ϕ_r(k)b_r(k) pour ϕ_i ∈ C (k).

2. Extraire les coefficients de F_{k+1} − F_k − f_k sur les b_i et obtenir un système de relations de récurrences du premier ordre en les ϕ_i.

3. Résoudre en les solutions rationnelles.

4. Si des ϕ_i sont trouvées, les renvoyer ; sinon, on a une preuve que de telles ϕ_i n’existent pas.

Exemples typiques par cette méthode n∑ k=0 (±1)^k(k + α)C^{(α)}_k (x) = (±1)^n 2(1 ∓ x) ( (n + 2α)C^{(α)}_n (x) ∓ (n + 1)C^{(α)}_{n+1}(x) )

(C^{(α)}_k (x) = polynômes ultrasphériques), n∑ k=1 (−1)^k (_m k ) H_k =

(−1)^n (_m n ) ( n + 1 m + 2

H_n + m + 1 − n (m + 2)^2 ) − m + 1 (m + 2)^2 , n∑ k=1 H^3_k = (n + 1)H^3_n − 3 2 (2n + 1)H^2_n + 3(2n + 1)H_n + 1 2 H^{(2)}_n − 6n ( H_n = 1 + 1 4 + 1 9 + · · · + 1 n^2 ) .

Algorithme de Zeilberger étendu

Entrée : un espace vectoriel V de base ( b_1(n, k), . . . , b_r(n, k) ) stable par les décalages en n et k ; un terme ∂-fini f_{n,k} de V . Sortie : des fractions rationnelles η_0(n), . . . , η_d(n), ϕ_1(n, k), . . . , ϕ_r(n, k) pour d minimal telles que η_d(n)f_{n+d,k} + . . . η_0(n)f_{n,k} = g_{n,k+1} − g_{n,k}, pour g_{n,k} = ϕ_1(n, k)b_1(n, k) + · · · + ϕ_r(n, k)b_r(n, k). Terminaison non garantie en général ; l’est dans le cas holonome.

Algorithme : Se ramener à une variante paramétrée de l’algorithme de Gosper étendu.

Exemples typiques par cette méthode ∞∑ n=0 r_n(a, b; q)z^n (q; q)_n = 1 (az; q)_∞(bz; q)_∞ (r_n = Rogers–Szegő) ∞∑ n=0 J_{2n+1/2}(x) = ∫_x 0 cos t √ 2πt dt n∑ k=0 ( n k )_2( n + k k )_2 = n∑ k=0 ( n k )( n + k k )_k∑ j=0 ( k j )_3 (van der Poorten) n∑ k=0 (_k∑ j=0 ( n j ))_3 = n 2 8^n + 8^n − 3n 4 2^n ( 2n n ) (Calkin) n∑ k=0 n+1−k ∑ l=0 l n + 1 ( n + 1 k )( n + 1 k + l ) = ( 2n n ) (Essam et Guttmann)

Non minimalité des opérateurs calculés n∑ k=0 (−1)^k ( n k )( pk n ) = (−p)^n (ordre p − 1 au lieu de 1), n∑ k=0 1 pk + 1 ( pk + 1 k )( p(n − k) n − k ) = ( pn + 1 n )

(ordre 2 au lieu de 1 pour p = 3), n∑ k=−n (−1)^kq^{(5k2−k)/2} (q; q)_{n−k}(q; q)_{n+k} (creative symmetrizing ; ordre 5 au lieu de 2).

η_d(n)u_{n+d,k} + · · · + η_0(n)u_{n,k} = v_{n,k+1} − v_{n,k} L’existence de v est une contrainte sur d !

Cette relation ne tient pas compte du lieu de sommation !

Programmes disponibles

• Package SumTools:-Hypergeometric de Maple8

• Bibliothèque Maple algolib du Projet Algo (Inria) disponible à l’URL http://algo.inria.fr/libraries/software.html

• Programmes Mathematica (et un peu de Maxima) du Risc à l’URL

http://www.risc.uni-linz.ac.at/research/combinat/

• Autre collection de programmes Maple et Mathematica disponible sur la page web du livre A = B à l’URL

http://www.cis.upenn.edu/~wilf/progs.html

• Programmes Maple de Zeilberger

http://www.math.rutgers.edu/~zeilberg/programs.html

Disponibilité des algorithmes

Sommation de séries génératrices :

• gfun (Murray, Salvy et Zimmermann), diffusé dans algolib (Maple).

• GeneratingFunctions (Mallinger), diffusé par le Risc (Mathematica).

Méthode de Fasenmyer :

• MultiSum (Wegschaider) et qMultiSum (Riese), diffusé par le Risc (Mathematica).

Algorithmes de Gosper et de Zeilberger :

• SumTools:-Hypergeometric (Koepf ; Le) dans Maple8. • fastZeil (Paule et Schorn), diffusé par le Risc (Mathematica).

• Zeilberger (Caruso), diffusé par le Risc (Maxima). • EKHAD (Zeilberger), (Maple).

q-Analogues des algorithmes de Gosper et de Zeilberger :

• qZeil (Riese), Bibasic Telescope (Riese), diffusé par le Risc (Mathematica).

• qEKHAD (Zeilberger), (Maple).

Extensions ∂-finies des algorithmes de Gosper et de Zeilberger :

• Mgfun (Chyzak), diffusé dans algolib (Maple).

Références

Ouvrages panoramiques :

• A = B, Marko Petkovšek, Herbert S. Wilf et Doron Zeilberger (1996).

Peters, Wellesley, MA.

• Chapitre Computer algebra, Frédéric Chyzak et Peter Paule. Dans :

Digital Library of Mathematical Functions, NIST. En préparation.

Sur la sommation rationnelle :

• The summation of rational functions, S. A. Abramov (1971). Zhurnal Vychislitelnoi Matematiki i Matematicheskoi Fisiki, (11), 1071–1075. Traduction dans Comput. Math. Math. Phys.

• Rational summation and Gosper-Petkovšek representation, Roberto Pirastu et Volker Strehl (1995). Journal of Symbolic Computation (20)5-6, 617–635.

Sur l’algorithme de Gosper et ses q-analogues :

• Decision procedure for indefinite hypergeometric summation, R.

William Gosper, Jr. (1978). Proceedings of the National Academy of Sciences of the USA (75)1, 40–42.

• Concrete mathematics, Ronald L. Graham, Donald E. Knuth et Oren Patashnik (1994). Deuxième édition. Addison-Wesley, Reading, MA.

• A generalization of Gosper’s algorithm to bibasic hypergeometric summation, Axel Riese (1996). Electronic Journal of Combinatorics (3)1, R19. 16 pages.

Sur la méthode de Fasenmyer :

• Some generalized hypergeometric polynomials, Mary Celine Fasenmyer (1945). Thèse. University of Michigan.

• Some generalized hypergeometric polynomials, Mary Celine Fasenmyer (1947). Bulletin of the American Mathematical Society (53), 806–812.

• A note on pure recurrence relations, Mary Celine Fasenmyer (1949).

American Mathematical Monthly 56, 14–17.

• Computer generated proofs of binomial multi-sum identities, Kurt Wegschaider (1997). Diplomarbeit. RISC, Université J. Kepler, Linz.

• qMultiSum — A package for proving q-hypergeometric multiple summation identities, Axel Riese. À paraître.

Sur l’algorithme de Zeilberger et ses q-analogues :

• A fast algorithm for proving terminating hypergeometric identities, Doron Zeilberger (1990). Discrete Mathematics (80)2, 207–211.

• The method of creative telescoping, Doron Zeilberger (1991). Journal of Symbolic Computation (11)3, 195–204.

• A Mathematica version of Zeilberger’s algorithm for proving binomial coefficient identities, Peter Paule et Markus Schorn (1995). Journal of Symbolic Computation, (20)5-6, 673–698.

• Greatest factorial factorization and symbolic summation, Peter Paule (1996). Journal of Symbolic Computation, (20)3, 235–268.

• A Mathematica q-analogue of Zeilberger’s algorithm based on an

algebraically motivated approach to q-hypergeometric telescoping, Peter Paule et Axel Riese (1997). Dans : Special Functions, q-Series and Related Topics (Toronto, ON, 1995), Fields Institute Communications (14). AMS, Providence, RI. p. 179–210.

• Applicability of Zeilberger’s algorithm to hypergeometric terms, S. A.

Abramov (2002). Dans : ISSAC’02 (July 7–10, 2002. Lille, France). Sur la sommation ∂-finie/holonome :

• A holonomic systems approach to special functions identities, Doron Zeilberger (1990). Journal of Computational and Applied Mathematics (32)3, 321–368.

• Démonstration “automatique” d’identités et fonctions

hypergéométriques (d’après D. Zeilberger), Pierre Cartier (1992). Astérisque (1991/92)206. Exp. No. 746, 3, 41–91.

• Non-commutative elimination in Ore algebras proves multivariate identities, Frédéric Chyzak et Bruno Salvy (1998). Journal of Symbolic

Computation (26)2, 187–227.

• Fonctions holonomes en calcul formel, Frédéric Chyzak (1998). Thèse.

INRIA, TU 0531.

• An extension of Zeilberger’s fast algorithm to general holonomic functions, F. Chyzak (2000). Discrete Mathematics (217)1-3, 115–134.

Sur la sommation dans des tours de corps à différences :

• Summation in finite terms, Michael Karr (1981). Journal of the ACM (28)2, 305–350.

• Theory of summation in finite terms, Michael Karr (1985). Journal of Symbolic Computation (1)3, 303–315.

• Symbolic summation in difference fields, Carsten Schneider (2001).

Thèse. RISC, Université J. Kepler, Linz. Autres références :

• Differentiably finite power series, R. P. Stanley (1980). European

Journal of Combinatorics (1)2, 175–188.

• D-finite power series, L. Lipshitz (1989). Journal of Algebra (122)2, 353–373.

• Rational functions certify combinatorial identities, Herbert S. Wilf et Doron Zeilberger (1990). Journal of the American Mathematical Society (3)1, 147–158.

• Rational function certification of multisum/integral/“q” identities, Herbert S. Wilf et Doron Zeilberger (1992). Bulletin of the American Mathematical Society (27)1, 148–153.

• Gfun: a Maple package for the manipulation of generating and holonomic functions in one variable, Bruno Salvy et Paul Zimmermann (1994). ACM Transactions on Mathematical Software (20)2, 163–177.

• An algorithm for finding recurrence relations of binomial sums and its complexity, Nobuki Takayama (1995). Journal of Symbolic Computation (20)5-6, 637–651.